

software

APPLE

Adventure per tutti...

di Guglielmo Nigri - Montesilvano (PE)

Tutti i possessori di computer si sono inbattuti, prima o poi, in un'adventure, versione computerizzata della vecchia caccia al tesoro, con in più l'enorme possibilità offerta dalla macchina alla fantasia dell'autore, e si sarà dibattuto inutilmente per ore tra paduli e pantani o avrà cercato inutilmente la strada nella nebbia più fitta per intere giornate. E a tutti sarà venuta prima o poi la tentazione di scriverne uno, ma avrà poi rinunciato vista la gran mole di lavoro necessaria alla realizzazione con in più lo svantaggio di non poterlo poi nemmeno giocare (a meno di lavaggi del cervello).

Due programmi che presentiamo permettono invece di costruire lo scenario e le regole di qualsiasi adventure e poi di giocarlo, naturalmente anche così l'autore non si divertirà molto, ma si può iniziare uno scambio di adventure tra vari autori a patto che tutti usino lo stesso programma. Proponiamo anzi agli autori di scambiare i dischetti con il testo degli adventure tramite le nostre rubriche micromarket e micrometing.

Capita spesso che molti abbiano buone idee su un gioco ma non sappiano come realizzarlo.

Quando ho fatto questa riflessione mi sono anche chiesto se non fosse possibile spiegare al computer lo svolgimento del gioco e lasciare a lui l'onore (e l'onere) di scriverlo.

Detto fatto ho realizzato questi programmi.

Il primo serve a spiegare il funzionamento del gioco alla mela rispondendo ad opportune domande.

Il secondo serve a giocare l'avventura utilizzando i dati salvati su memoria di massa dal primo (può sembrare complicato ma non lo è). ADVENTUREDITOR può salvare su disco adventure game anche molto estesi, fino a 100 stanze, 100 oggetti e un dizionario di 150 parole oltre ad innumerevoli altre cose.

Considerando che da ogni stanza ci si può muovere in sei direzioni, che per ogni verbo si può avere una risposta personalizzata, spostamenti di oggetti, esecuzione fattibile solo una volta o solo in caso che si possieda un certo oggetto o che ci si trovi in

un determinato luogo si può finalmente pensare che non si possa fare di meglio.

Esaminiamo il modus operandi nel caso si voglia scrivere un'avventura.

Bisogna naturalmente annotarsi tutto ma proprio tutto dell'adventure che si ha in mente (improvvisare davanti al computer non è cosa facile).

Dato il RUN al programma ADVENTUREDITOR, l'Apple chiederà in primo luogo il nome del gioco e il numero di luoghi del gioco (1-100). Verrà allora chiesta la descrizione della stanza numero 1, ciò che si scriverà apparirà esattamente nello stesso modo durante il gioco (attenzione a non inserire virgole pena l'input error), si raccomanda piuttosto di non inserire in adventure molto estesi descrizioni più lunghe di 4 linee video (160 car.).

Di seguito verrà chiesto il numero della stanza che si può raggiungere dando il comando N (nord) durante il gioco e così via per le altre 5 direzioni.

Quindi il computer chiederà: "Va bene?", rispondendo NO l'Apple richiederà il tutto, rispondendo SI il computer proseguirà a far domande.

A\$	= VERBO DATO.	OG\$()	= NOMI DEGLI OGGETTI.
AN\$	= NOME DELL' ADVENTURE.	PO()	= POSIZIONI DEGLI OGGETTI.
AP()	= AUMENTO PUNTEGGIO RELATIVO AL VERBO.	PP	= CONTATORE OGGETTI POSSEDUTI.
B\$	= COMPLEMENTO DATO.	PR\$	= PRESENTAZIONE.
CO\$()	= COMPLEMENTI DEI VERBI PROGRAMMATI.	PUO()	= LUOGHI DOVE E' POSSIBILE ESEGUIRE I VERBI.
D\$	= CHR\$(4)	SC	= PUNTEGGIO.
D(A,B)	= DIREZIONE B DALLA STANZA A.	SET	= POSIZIONE ATTUALE DEL GIOCATORE.
ER\$	= MESSAGGIO DI ERRORE.	ST\$()	= DESCRIZIONI LUOGHI.
EX()	= FLAG DI VERBO GIA' ESEGUITO.	TF	= FLAG DI PRESENZA PUNTEGGIO.
FF.	= FLAG GENERICO.	TN	= PUNTEGGIO DA RAGGIUNGERE.
FL()	= FLAG DI VERBO NON ESEGUIBILE PIU' VOLTE.	V\$()	= VERBI.
FO()	= SPOSTAMENTI DI OGGETTI.	VPO()	= COPIA DEL VETTORE PO()
H\$	= STRINGA GENERICA.	VSET	= COPIA DI SET INIZIALE.
I, I2	= FOR-NEXT.	W\$	= FRASE INSERITA.
ME\$()	= MESSAGGI DEI VERBI.	X\$	= STRINGA GENERICA.
NO	= NUMERO DEGLI OGGETTI.	X2\$	= " " " "
NS	= NUMERO DEI LUOGHI.	Y	= NUMERO ERRORE.
NV	= NUMERO DEI VERBI.	ZO()	= LUOGHI DA RAGGIUNGERE ESEGUITO UN COMANDO.
ODA()	= OGGETTI DA AVERE PER ESEGUIRE I VERBI.		

Figura 1 - Elenco delle variabili usate nel programma.

```

100 NOTRACE
110 PRINT CHR$(4)"NONON C,I,0": VTAB PEEK(37): CALL - 868
120 PRINT CHR$(4)"MAXFILES!"
130 TEXT : HOME : SPEED= 255
140 DIM ST$(100),OG$(100),PO(100),D(100,6),V$(75),CO$(75),ME$(75),ZO(75)
    ,FO(75),PUO(75),ODA(75),AP(75),FL(75)
150 DS = CHR$(4)
160 PRINT "*ADVENTUREEDITOR By Guglielmo Nigri 1984*": POKE 34,2
170 PRINT "Scrivi il nome dell' avventura."
180 INPUT "":IAN$
190 IF AN$ = "" THEN 180
200 GOSUB 1090: IF FF THEN 170
210 POKE 49168,0
220 TF = 1:TN = 1
230 PRINT : PRINT "1' avventura e' basata sul punteggio?": GET H$: PRINT
240 IF H$ = "N" THEN TF = 0: GOTO 260
250 INPUT "Che punteggio si deve raggiungere ?":TN: GOSUB 1090: IF FF THEN
260 PRINT : PRINT "Quanti luoghi ci sono nel gioco ?"
270 INPUT "":I:MX = 100: GOSUB 1130: IF FF THEN 270
280 NS = I
290 FOR I = 1 TO NS
300 PRINT : PRINT "Scrivi la descrizione del luogo N.":I
310 INPUT "":ST$(I)
320 PRINT "Che luogo raggiungo andando verso..."
330 INPUT "NORD (N) ?":D(I,1)
340 INPUT "SUD (S) ?":D(I,2)
350 INPUT "EST (E) ?":D(I,3)
360 INPUT "OVEST (O) ?":D(I,4)
370 INPUT "SU (SALI) ?":D(I,5)
380 INPUT "GIU' (SCENDI)?":D(I,6)
390 GOSUB 1090: IF FF THEN 300
400 NEXT
410 PRINT : PRINT "Quanti oggetti ci sono nel gioco ?"
420 INPUT "":I:MX = 100: GOSUB 1130: IF FF THEN 420
430 NO = I
440 FOR I = 1 TO NO
450 PRINT : PRINT "Scrivi il nome dell' oggetto N.":I
460 INPUT "":OG$(I)
470 PRINT : PRINT "In che luogo si trova ?"
480 INPUT "":PO(I)
490 GOSUB 1090: IF FF THEN 450
500 NEXT
510 PRINT : PRINT "Quanti sono i verbi speciali?"
520 INPUT "":I:MX = 75: GOSUB 1130: IF FF THEN 520
530 NV = I
540 FOR I = 1 TO NV
550 PRINT : PRINT "Scrivi l' imperativo del verbo N.":I
560 INPUT "":V$(I)
570 IF V$(I) = "" THEN 560
580 PRINT : PRINT "Scrivi il suo complemento."
590 INPUT "":CO$(I)
600 PRINT : PRINT "Scrivi la risposta da dare al giocatore."
610 INPUT "":ME$(I)
620 IF ME$(I) = "" THEN ME$(I) = "Ok.": PRINT "Ok."
630 PRINT : PRINT "In che luogo vado (0 = Sto fermo)"
640 INPUT "":ZO(I)
650 PRINT : PRINT "Inserisci lo spostamento di oggetti."
660 INPUT "":FO(I)
670 IF NOT TF THEN 700
680 PRINT : PRINT "Effetto sul punteggio (+n,-n,0) ?"
690 INPUT "":AP(I)
700 PRINT : PRINT "Dove posso eseguire il comando ?"
710 INPUT "":PUO(I)
720 PRINT : PRINT "Che oggetto devo avere con me ?"
730 INPUT "":ODA(I)
740 PRINT
750 HTAB 1: POKE 49168,0: PRINT "Posso farlo piu' di una volta (S/N)
?": GET H$: IF H$ = "N" THEN FL(I) = 1: GOTO 770
760 IF H$ < "S" THEN 750
770 GOSUB 1090: IF FF THEN 550
780 NEXT
790 PRINT : PRINT "Scrivi la presentazione del gioco"
800 INPUT "":PR$
810 GOSUB 1090: IF FF THEN 790
820 PRINT : INPUT "Da che luogo si inizia ?":SET
830 GOSUB 1090: IF FF THEN 820
840 PRINT CHR$(7)
850 PRINT "INSERISCI IL DISCO E PREMI RETURN ": GET H$: PRINT
860 PRINT D$"OPEN"AN$
870 PRINT D$"DELETE"AN$
880 PRINT D$"OPEN"AN$
890 PRINT D$"WRITE"AN$
900 PRINT PR$: PRINT NS: PRINT NO: PRINT NV: PRINT TF: PRINT TN: PRINT
SET
910 FOR I = 1 TO NS
920 PRINT ST$(I)
930 FOR I2 = 1 TO 6
940 PRINT D(I,I2)
950 NEXT
960 NEXT
970 FOR I = 1 TO NO
980 PRINT OG$(I): PRINT PO(I)
990 NEXT
1000 FOR I = 1 TO NV
1010 PRINT V$(I): PRINT CO$(I): PRINT ME$(I)
1020 PRINT ZO(I): PRINT FO(I): PRINT ODA(I): PRINT AP(I)
: PRINT FL(I)
1030 NEXT
1040 PRINT D$"CLOSE"AN$
1050 POKE 49168,0
1060 PRINT : PRINT "Vuoi scrivere un' altra avventura ?": GET H$
1070 IF H$ = "S" THEN RUN
1080 TEXT : HOME : NEW
1090 FF = FRE(0):FF = 0: PRINT CHR$(7): INPUT "VA BENE ?":H$
1100 IF H$ = "" THEN 1090
1110 IF ASC (H$) < > 83 THEN FF = 1
1120 RETURN
1130 FF = 0: IF I < 1 OR I > MX THEN FF = 1: PRINT CHR$(7):"VALORE NON
AMMESSO, RIPROVA !"
1140 RETURN
1150 END

```

Listato 1 - Programma EDITOR, permette di creare un' avventura e scrive poi in un file di testo tutte le informazioni necessarie al gioco.

Finiti i luoghi il calcolatore chiederà il numero di oggetti presenti nell'avventure.

Di ogni oggetto verrà chiesto il nome (da scrivere *senza* articolo) e la posizione iniziale.

Successivamente si passerà ai verbi. Ogni verbo presuppone un suo complemento. Il dizionario massimo è di 150 parole suddivise in 75 verbi e altrettanti complementi.

Per prima cosa l'Apple chiederà il numero dei verbi, quindi di ogni verbo l'imperativo e il suo complemento (sempre *senza* articolo). La domanda successiva sarà inerente alla risposta da dare nel caso sia possibile eseguire il comando, il default dato premendo ENTER è "OK.". Poi verrà chiesto il luogo da raggiungere impartito il comando, scrivendo 0 non ci si muoverà.

Verrà chiesto in seguito lo spostamento di oggetti da effettuare. Inserendo un numero positivo, ad esempio +5, l'oggetto numero 5 apparirà magicamente nel luogo dove ci si trova. Inserendo un numero negativo, ad esempio -3, l'oggetto 3 scomparirà misteriosamente (in realtà la sua posizione sarà data a 0, la stanza 0 è un luogo che non c'è e quindi è utile per far sparire oggetti o depositarli momentaneamente).

Inserendo 0 si intende che non deve avvenire alcuno spostamento. L'amica mela chiederà poi l'eventuale effetto sul punteggio (se il gioco è basato su di esso). Sarà successivamente chiesto il luogo dove è possibile eseguire il comando. Lo 0 indica che il comando è eseguibile dappertutto. Un numero negativo, ad esempio -7, indica che nella stanza n. 7 non potrà essere impartito quel comando. Apple seguirà chiedendo che oggetto si deve avere per eseguire l'ordine, scrivendo 0 si intende che non è necessario alcun oggetto. Verrà poi chiesto se il comando è eseguibile più di una volta, rispondere premendo S o N.

Le ultime cose chieste saranno il numero della stanza da dove inizia il gioco e la presentazione. Verranno quindi date le istruzioni relative al salvataggio su disco dei dati. Purtroppo i dati sono molti, ma da parte mia ho fatto di tutto per ridurre l'ingombro su disco.

Questo programma è disponibile su disco presso la redazione. Vedere l'elenco dei programmi disponibili e le istruzioni per l'acquisto a pag. 153.

Veniamo al secondo programma:

ADVENTURE

Come già detto esso serve a giocare l'avventura programmata usando l'editor.

Dato il RUN verrà chiesto il nome dell'avventura da caricare. Caricata, si inizierà immediatamente a giocare. Per ogni stanza verranno visualizzati i nomi degli oggetti presenti insieme alla descrizione e verrà atteso un comando nella parte inferiore del video.

Infatti, usando opportunamente le finestre di testo, nella parte superiore del video è costantemente visualizzato ciò che è possibile vedere, mentre in quella inferiore avviene il dialogo tra giocatore e computer. Ci sono molti verbi assegnati "d'ufficio" e che non hanno bisogno di essere programmati, essi sono:

I, INVENTARIO, N, S, E, O, SALI, SCENDI, PRENDI, PIGLIA, POSA, LASCIA, BUTTA, SALVA GIOCO, PUNTEGGIO.

I verbi eseguiti per primi sono quelli programmati dall'utente, quindi nessuno vieta di riprogrammare quelli originari. Una cosa molto bella e utile è che l'articolo può esserci o non esserci e può essere di qualsiasi lunghezza.

Si potrà quindi dire ACCENDI TORCIA oppure ACCENDI LA TORCIA o anche ACCENDI CODESTA TORCIA.

In questa versione, infatti, la prima parola viene considerata verbo e l'ultima complemento. Questo è anche il motivo per cui i nomi degli oggetti devono essere dati senza l'articolo ad ADVENTUREDITOR.

Ritengo ADVENTUREEDITOR un programma semplice da leggere, quindi passo subito a commentare ADVENTURE che è un programma più lungo e con alcune cose da notare. La numerazione è da 100 con passo 10. Le prime istruzioni servono per eliminare l'effetto del MON e del TRACE.

Le linee da 100 a 190 sono relative all'inizializzazione e alla presentazione. Da 200 a 320 si stampa la descrizione della stanza. Da 330 a 520 si accetta il comando e si assegnano ad A\$ e a B\$ rispettivamente il verbo e il complemento. Dalla linea 530 alla 710 c'è la routine di riconoscimento e di esecuzione dei verbi programmati dall'utente, mentre tra 720 e 1390 c'è quella relativa ai comandi preprogrammati. La linea 1400 segnala un eventuale errore di battitura. A 1410 c'è la routine di morte

Adventure Player

```

100 NOTRACE : TEXT : HOME : SPEED= 255
110 PRINT CHR$(4)"NOMON C,I,O": VTAB PEEK (37): CALL - 868
120 PRINT CHR$(4)"MAXFILES1"
130 D$ = CHR$(4):ER$ = "Non lo posso ancora fare !"
140 HOME : PRINT "* ADVENTURE By Guglielmo Nigri 1984 *"
150 VTAB 8: PRINT "A che avventura vuoi giocare ?"
160 INPUT ">":AN$: IF AN$ = "" THEN 160
170 GOSUB 1540
180 HOME
190 VTAB 18: PRINT PR$,
200 IF SET = - 2 THEN 1410
210 IF SET = - 3 THEN 1520
220 POKE 34,0: POKE 35,12: HOME : PRINT ST$(SET)
230 FF = 0
240 FOR I = 1 TO NV
250 IF PD(I) < > SET THEN 300
260 IF FF THEN 290
270 PRINT "Gli oggetti visibili sono : "
280 FF = 1
290 PRINT OG$(I),
300 NEXT
310 POKE 34,12: POKE 35,24
320 PRINT : VTAB 24
330 PRINT "Cosa devo fare ?"
340 INPUT "":W$
350 IF W$ = "" THEN 340
360 PRINT
370 IF ASC ( RIGHT$( W$,1)) < 65 THEN W$ = LEFT$( W$, LEN (W$) - 1)
380 X$ = W$: GOSUB 1990: IF FF THEN 420
390 A$ = W$
400 B$ = ""
410 GOTO 530
420 A$ = "": I = 0
430 I = I + 1
440 X$ = MID$( W$,I,1)
450 IF X$ = " " THEN 470
460 A$ = A$ + X$: GOTO 430
470 B$ = "": I = LEN (W$) + 1
480 I = I - 1
490 X$ = MID$( W$,I,1)
500 IF X$ = " " THEN 530
510 B$ = X$ + B$
520 GOTO 480
530 FF = 0
540 FOR I = 1 TO NV
550 IF A$ < > V$(I) THEN 690
560 IF B$ < > CD$(I) THEN 690
570 IF FL(I) AND EX(I) THEN PRINT "L' ho già fatto !": GOTO 680
580 IF ODA(I) THEN IF PD(ODA(I)) < > - I THEN PRINT ER$: GOTO 680
590 IF PUO(I) = 0 THEN 620
600 IF SGN (PUO(I)) = - 1 THEN IF SET = ABS (PUO(I)) THEN PRINT
"Non posso farlo qui !": GOTO 680
610 IF PUO(I) < > SET THEN PRINT ER$: GOTO 680
620 PRINT ME$(I)
630 IF ZO(I) THEN SET = ZO(I)
640 IF FO(I) = 0 THEN 670
650 IF SGN (FO(I)) = 1 THEN PD(FO(I)) = SET: GOTO 670
660 PD( ABS (FO(I))) = 0
670 SC = SC + AP(I):EX(I) = 1
680 FF = 1:I = NV
690 NEXT
700 IF SC > = TN THEN 1520
710 IF FF THEN 200
720 IF NOT (A$ = "PUNTEGGIO" AND TF) THEN 740
730 PRINT "Finora ho totalizzato "ISC:" punti.": GOTO 330
740 IF NOT (A$ = "PIGLIA" OR A$ = "PRENDI") THEN 870
750 IF B$ = "" THEN PRINT "Se mi dici cosa...": GOTO 330
760 IF PP = 8 THEN PRINT "Non posso, sono troppo carico !": GOTO 330
770 FF = 0
780 FOR I = 1 TO NV
790 IF NOT (B$ = OG$(I) AND PD(I) = SET) THEN 840
800 PRINT "OK."
810 FF = 1
820 PD(I) = - 1
830 PP = PP + 1
840 NEXT
850 IF NOT FF THEN PRINT B$:" ? Non ne vedo qui intorno...": GOTO 330
860 GOTO 200
870 IF NOT (A$ = "POSA" OR A$ = "LASCIA" OR A$ = "BUTTA") THEN 980
880 FF = 0
890 FOR I = 1 TO NV
900 IF NOT (B$ = OG$(I) AND PD(I) = - 1) THEN 950
910 PRINT "OK."
920 FF = 1
930 PD(I) = SET
940 PP = PP - 1
950 NEXT
960 IF FF THEN 200
970 PRINT "Non ho "B$:" con me !": GOTO 330
980 IF NOT (A$ = "SALVA" AND B$ = "GIOCO") THEN 1130
990 HOME
1000 PRINT D$"OPEN"AN$.GD"
1010 PRINT D$"DELETE"AN$.GD"
1020 PRINT D$"OPEN"AN$.GD"
1030 PRINT D$"WRITE"AN$.GD"
1040 PRINT SET: PRINT PP: PRINT SC
1050 FOR I = 1 TO NV
1060 PRINT PD(I)
1070 NEXT
1080 FOR I = 1 TO NV
1090 PRINT EX(I)

```

```

1100 NEXT
1110 PRINT D*"CLOSE"AN$.GD"
1120 HOME : GOTO 330
1130 IF A$ < > "BASTA" THEN 1160
1140 IF NOT IF THEN PRINT : GOTO 1420
1150 PRINT "In questa avventura ho totalizzato ": PRINT SC:" punti cioe'
      11 "SC / 100 * TN" % del totale.": PRINT : GOTO 1420
1160 IF NOT (A$ = "I" OR A$ = "INVENTARIO") THEN 1270
1170 FF = 0
1180 FOR I = 1 TO NO
1190 IF PO(I) < > - 1 THEN 1240
1200 IF FF THEN 1230
1210 PRINT "Ora con me ho : "
1220 FF = 1
1230 PRINT TAB( 4);OG$(I)
1240 NEXT
1250 IF FF THEN 330
1260 PRINT "Ora non ho niente con me...": GOTO 330
1270 IF NOT ((A$ = "N" AND D(SET,1)) OR (A$ = "VAI" AND B$ = "NORD" AND
D(SET,1))) THEN 1290
1280 SET = D(SET,1): GOTO 200
1290 IF NOT ((A$ = "S" AND D(SET,2)) OR (A$ = "VAI" AND B$ = "SUD" AND
D(SET,2))) THEN 1310
1300 SET = D(SET,2): GOTO 200
1310 IF NOT ((A$ = "E" AND D(SET,3)) OR (A$ = "VAI" AND B$ = "EST" AND
D(SET,3))) THEN 1330
1320 SET = D(SET,3): GOTO 200
1330 IF NOT ((A$ = "O" AND D(SET,4)) OR (A$ = "VAI" AND B$ = "OVEST" AND
D(SET,4))) THEN 1350
1340 SET = D(SET,4): GOTO 200
1350 IF NOT (A$ = "SALI" AND D(SET,5)) THEN 1370
1360 SET = D(SET,5): GOTO 200
1370 IF NOT (A$ = "SCENDI" AND D(SET,6)) THEN 1390
1380 SET = D(SET,6): GOTO 200
1390 IF LEN (A$) = 1 OR A$ = "SALI" OR A$ = "SCENDI" OR A$ = "VAI" THEN
PRINT "Non posso andare in quella direzione !": GOTO 330
1400 PRINT "Non ti capisco, spiegati meglio ": GOTO 330
1410 PRINT : PRINT "Hai perso miseramente, mi dispiace...": PRINT
1420 HTAB 1: PRINT "Giochi ancora ? ":
1430 GET H$: PRINT H$:
1440 IF H$ = "N" THEN TEXT : HOME : NEW
1450 IF H$ < > "S" THEN 1420
1460 FOR I = 1 TO NO
1470 PO(I) = VFO(I)
1480 NEXT I
1490 SET = VSET
1500 PP = 0
1510 GOSUB 1830: GOTO 180
1520 PRINT : PRINT "Congratulazioni, HAI VINTO !!!": PRINT
1530 GOTO 1420
1540 ONERR GOTO 1800
1550 PRINT D*"VERIFY"AN$
1560 PRINT D*"OPEN"AN$
1570 PRINT D*"READ"AN$
1580 INPUT PR$,NS,NO,NV,TF,TN,SET
1590 DIM ST$(NS),DG$(NO),PO(NO),D(NS,6),V$(NV),CO$(NV),ME$(NV),ZO(NV),FO
(NV),PUO(NV),ODA(NV),AP(NV),FL(NV),EX(NV)
1600 FOR I = 1 TO NS
1610 INPUT ST$(I)
1620 FOR I2 = 1 TO 6
1630 INPUT D(I,I2)
1640 NEXT
1650 NEXT
1660 FOR I = 1 TO NO
1670 INPUT OG$(I),PO(I)
1680 NEXT
1690 FOR I = 1 TO NV
1700 INPUT V$(I),CO$(I),ME$(I)
1710 INPUT ZO(I),FO(I),PUO(I),ODA(I),AP(I),FL(I)
1720 NEXT
1730 PRINT D*"CLOSE"AN$
1740 FOR I = 1 TO NO
1750 VFO(I) = PO(I)
1760 NEXT
1770 VSET = SET
1780 GOSUB 1830
1790 RETURN
1800 POKE 216,0:Y = PEEK (222): IF Y = 6 THEN PRINT : PRINT "Non esist
e su questo disco !": GOTO 160
IF Y = 11 THEN PRINT : PRINT "Il nome non e' valido !": GOTO 160
1810 RESUME
1820 PRINT : PRINT "Vuoi caricare una partita interrotta ?":
1830 HTAB 39
1840 GET H$: PRINT H$:
1850 IF H$ = "N" THEN RETURN
1860 IF H$ < > "S" THEN 1840
1870 PRINT D*"OPEN"AN$.GD"
1880 PRINT D*"READ"AN$.GD"
1890 INPUT SET,PP,SC
1900 FOR I = 1 TO NO
1910 INPUT PO(I)
1920 NEXT
1930 NEXT
1940 FOR I = 1 TO NV
1950 INPUT EX(I)
1960 NEXT
1970 PRINT D*"CLOSE"AN$.GD"
1980 RETURN
1990 FF = 0: FOR I = 1 TO LEN (X$)
2000 X2$ = MID$( X$,I,1)
2010 IF X2$ = " " THEN FF = 1
2020 NEXT
2030 RETURN

```

Listato 2 - Programma ADVENTURE, con questo programma si possono giocare gli adventure creati (possibile da altri) con l'EDITOR.

mentre a 1520 e 1530 c'è quella della vittoria.

Tra 1420 e 1510 vi sono le domande a conclusione del gioco. Tra 1540 e 1980 c'è tutta la parte relativa alla lettura di dati da disco. Da 1990 a 2030 c'è una routine che controlla l'esistenza di uno spazio nella stringa X\$.

Una caratteristica piacevole è che è possibile salvare la situazione momentanea in qualunque istante del gioco usando il comando SALVA GIOCO. È necessario ancora un piccolo chiarimento: se una stanza deve essere illuminata da un comando tipo ACCENDI LA TORCIA conviene avere due stanze, una con la descrizione del tipo "NON VEDO NIENTE." e un'altra con la stanza illuminata. Dando il comando ACCENDI LA TORCIA si dovrà passare in quell'ultima specificandone il numero alla richiesta di ADVENTUREEDITOR.

Nota: il file dell'adventure è un normale file di testo che può essere letto facilmente dal disco, quindi, anche se la tentazione è forte, cercate di resistere...

```

10 TEXT : HOME :D$ = CHR$( 4)
20 PRINT D*"OPEN PROVA"
30 PRINT D*"WRITE PROVA"
40 READ A$: IF A$ = "#" THEN 100

50 PRINT A$: GOTO 40
100 PRINT D*"CLOSE"
200 DATA "L' AVVENTURA SIA CON
TE '----- GUADAGNA
100 PUNTI E VINCERAI"
201 DATA 3,3,4,1,100,2
202 DATA "SONO IN UNA STANZA MO
LTO IMPRESSIONANTE.C'E' UNA
USCITA A NORD E UNA PORTA----
---CHIUSA A CHIAVE."
203 DATA 2,0,0,0,0,0
204 DATA "SONO IN UNO STRANO CO
RRIDIO PIENO-----DI RAGNAT
ELE CHE PORTA A SUD E AD EST
"
205 DATA 2,1,3,0,0,0
206 DATA "SONO IN UNA STANZA IL
LUMINATA TETRAMENTEDA UNA FI
NESTRA. L'USCITA E' AD OVEST
. C'E' UNA TEDA FISSATA PER
TERRA. E' DI-----UN VETRO MOLT
O SCURO CHE NON MI CONVINCER
ER NIENTE. C'E' ANCHE UNA ST
RANA LEVA_NEL MURO !!"
207 DATA 0,0,0,2,0,0
208 DATA "COMPUTER",1,"CHIAVE",
0,"MARTELLO",2,"ACCENDI","CO
MPUTER","SI. MI DIVERTO MOLT
O CON QUESTI GIOCHI !"
209 DATA 0,0,0,1,25,1
210 DATA "ROMPI","TECA","CRASH!
TINKLE! SI. VEDO QUALCOSA..
"
211 DATA 0,2,3,3,25,1
212 DATA "APRI","PORTA","BENE.
GUADAGNI 50 PUNTI !"
213 DATA 0,0,1,2,50,1
214 DATA "TIRA","LEVA","FATTO.
AAAAAGGGGHHH ERA ELETTRIFIC
ATA !!"
215 DATA -2,0,3,0,0,0
220 DATA "#"

```

Listato 3 - Semplice programmino per creare un Adventure di prova con sole tre stanze ma abbastanza difficile da risolvere (non barate!).

STAMPANTI Epson,

una scelta prestigiosa, senza compromessi

Scegliere una stampante è facile?

C'è una sola regola, pretendere sempre il massimo delle prestazioni, senza compromessi: materiali e componenti di prima qualità, disegno elegante, grande affidabilità, robustezza, facilità e flessibilità d'impiego, prezzo adeguato e la garanzia di un grande costruttore leader mondiale.

Così, con Epson, la scelta è facile e sicura.

Epson il più grande costruttore al mondo di stampanti vi offre una gamma di prodotti prestigiosi che soddisfano ogni vostra necessità. Epson, una soluzione raffinata, in esclusiva per il vostro computer.



FX-80


Indispensabile nelle applicazioni in cui la versatilità e la qualità di stampa sono un imperativo.

Possibilità di creare qualsiasi carattere su una matrice di 11x9 punti. Memoria RAM da 4 Kbyte. 256 tipi di caratteri definibili dall'utente. 136 tipi di caratteri a corredo. Alta velocità di stampa a 160 caratteri al secondo su 80 colonne. Ben 9 modi di stampa grafica punto a punto selezionabili sulla stessa riga contemporaneamente.

Inseritori automatici di fogli singoli a singola e doppia vaschetta.

FX-100

Con 132 colonne e 160 caratteri al secondo, la FX-100 è la stampante ideale per data processing e tabulati, specialmente in ufficio grazie anche alla possibilità di accettare fogli singoli e moduli continui di qualsiasi formato. La matrice di punti 11x9 consente prestazioni grafiche e la formazione di una grandissima varietà di tipi di carattere, fino a 256, definibili anche dall'utente e memorizzati nei 3 kbyte di RAM interna. La FX-100 non teme la fatica: la testina di stampa è garantita per oltre 100 milioni di caratteri ed è facilmente sostituibile. Inseritori automatici di fogli singoli.



RX-80 / RX-100

Le migliori prestazioni da stampanti, semplici, versatili, silenziose e veloci con 100 caratteri al secondo. 128 tipi di caratteri selezionabili e 11 set internazionali. 80 o 132 colonne. 6 diverse possibilità grafiche. Tutti i tipi di carta, modulo continuo, foglio singolo. Inseritori automatici di fogli singoli.



studio martinetti

Epson dunque, senza compromessi.

EPSON

EPSON CORPORATION
HEAD OFFICE
80 Hirooka, Shiojiri-shi, Nagano
399-07 JAPAN

EPSON, computer e periferiche
sono prodotti distribuiti,
assistiti e garantiti
da SEGI S.p.A. - Milano Via Timavo, 12

segi SERVIZI
GENERALI PER
L'INFORMATICA

